Expressions

```
Head[{1, 2, 3}]
```

```
List
```

```
FullForm[x + y]
```

```
Plus[x, y]
```

```
Head[x]
```

```
Symbol
```

```
FullForm[2]
```

```
2
```

```
Head[2]
```

```
Integer
```

```
ToString[2]
```

```
2
```

```
FullForm[%]
```

```
"2"
```

```
TreeForm[a - 2 b^2]
```



```
SymmetricPolynomial[2, Symbol /@ CharacterRange["a", "z"]]
```

$$\begin{aligned}
&a\,b + a\,c + b\,c + a\,d + b\,d + c\,d + a\,e + b\,e + c\,e + d\,e + a\,f + b\,f + c\,f + d\,f + e\,f + a\,g + b\,g + c\,g + d\,g + e\,g + f\,g + \\
&\ a\,h + b\,h + c\,h + d\,h + e\,h + f\,h + g\,h + a\,i + b\,i + c\,i + d\,i + e\,i + f\,i + g\,i + h\,i + a\,j + b\,j + c\,j + d\,j + \\
&\ e\,j + f\,j + g\,j + h\,j + i\,j + a\,k + b\,k + c\,k + d\,k + e\,k + f\,k + g\,k + h\,k + i\,k + j\,k + a\,l + b\,l + c\,l + d\,l + \\
&\ e\,l + f\,l + g\,l + h\,l + i\,l + j\,l + k\,l + a\,m + b\,m + c\,m + d\,m + e\,m + f\,m + g\,m + h\,m + i\,m + j\,m + k\,m + l\,m + \\
&\ a\,n + b\,n + c\,n + d\,n + e\,n + f\,n + g\,n + h\,n + i\,n + j\,n + k\,n + l\,n + m\,n + a\,o + b\,o + c\,o + d\,o + e\,o + f\,o + \\
&\ g\,o + h\,o + i\,o + j\,o + k\,o + l\,o + m\,o + n\,o + a\,p + b\,p + c\,p + d\,p + e\,p + f\,p + g\,p + h\,p + i\,p + j\,p + k\,p + \\
&\ l\,p + m\,p + n\,p + o\,p + a\,q + b\,q + c\,q + d\,q + e\,q + f\,q + g\,q + h\,q + i\,q + j\,q + k\,q + l\,q + m\,q + n\,q + o\,q + \\
&\ p\,q + a\,r + b\,r + c\,r + d\,r + e\,r + f\,r + g\,r + h\,r + i\,r + j\,r + k\,r + l\,r + m\,r + n\,r + o\,r + p\,r + q\,r + a\,s + \\
&\ b\,s + c\,s + d\,s + e\,s + f\,s + g\,s + h\,s + i\,s + j\,s + k\,s + l\,s + m\,s + n\,s + o\,s + p\,s + q\,s + r\,s + a\,t + b\,t + \\
&\ c\,t + d\,t + e\,t + f\,t + g\,t + h\,t + i\,t + j\,t + k\,t + l\,t + m\,t + n\,t + o\,t + p\,t + q\,t + r\,t + s\,t + a\,u + b\,u + \\
&\ c\,u + d\,u + e\,u + f\,u + g\,u + h\,u + i\,u + j\,u + k\,u + l\,u + m\,u + n\,u + o\,u + p\,u + q\,u + r\,u + s\,u + t\,u + a\,v + \\
&\ b\,v + c\,v + d\,v + e\,v + f\,v + g\,v + h\,v + i\,v + j\,v + k\,v + l\,v + m\,v + n\,v + o\,v + p\,v + q\,v + r\,v + s\,v + t\,v + \\
&\ u\,v + a\,w + b\,w + c\,w + d\,w + e\,w + f\,w + g\,w + h\,w + i\,w + j\,w + k\,w + l\,w + m\,w + n\,w + o\,w + p\,w + q\,w + r\,w + \\
&\ s\,w + t\,w + u\,w + v\,w + a\,x + b\,x + c\,x + d\,x + e\,x + f\,x + g\,x + h\,x + i\,x + j\,x + k\,x + l\,x + m\,x + n\,x + o\,x + \\
&\ p\,x + q\,x + r\,x + s\,x + t\,x + u\,x + v\,x + w\,x + a\,y + b\,y + c\,y + d\,y + e\,y + f\,y + g\,y + h\,y + i\,y + j\,y + k\,y + \\
&\ l\,y + m\,y + n\,y + o\,y + p\,y + q\,y + r\,y + s\,y + t\,y + u\,y + v\,y + w\,y + x\,y + a\,z + b\,z + c\,z + d\,z + e\,z + f\,z + \\
&\ g\,z + h\,z + i\,z + j\,z + k\,z + l\,z + m\,z + n\,z + o\,z + p\,z + q\,z + r\,z + s\,z + t\,z + u\,z + v\,z + w\,z + x\,z + y\,z
\end{aligned}$$

In[10]:=
```
? Notebook
```

```
Notebook[{cell_1, cell_2, ...}] is the low-level construct that
   represents a notebook manipulated by the Mathematica front end.  >>
```

Quick Start

■ Equal signs

```
(a + b) ^ 2 = a^2 + 2 * a * b + b^2
```

```
Set::write : Tag Power in (a + b)^2 is Protected. >>
```

$a^2 + 2\,a\,b + b^2$

```
(a + b) ^ 2 == a^2 + 2 * a * b + b^2
```

$(a + b)^2 == a^2 + 2\,a\,b + b^2$

```
Simplify[%]
```

```
True
```

```
(a + b) ^ 2 === a^2 + 2 * a * b + b^2
```

```
False
```

In[11]:=
```
test := Print["this is a test"]
```

In[12]:=
```
test
```

```
this is a test
```

Programming constructs

```
If[2 > 3, "true", "false"]
```

```
false
```

```
If[x == y, "true", "false", "don't know"]
```

```
don't know
```

```
Do[Print[n], {n, 0, 3}]
```

```
0
```

```
1
```

```
2
```

```
3
```

Concepts

List operations

matrix-vector multiplication:

```
{{1, 2}, {0, 1}}.{2, -1}
```

```
{0, -1}
```

```
Table[2 * n, {n, 1, 10}]
```

```
{2, 4, 6, 8, 10, 12, 14, 16, 18, 20}
```

```
Range[10]^3
```

```
{1, 8, 27, 64, 125, 216, 343, 512, 729, 1000}
```

```
{1, 2, 3} * {7, 8, 9}
```

```
{7, 16, 27}
```

In[13]:=
```
{1, 2, 3}^{7, 8, 9}
```

Out[13]=
```
{1, 256, 19683}
```

```
MapThread[Append, {{{1, 2}, {0, 1}}, {2, -1}}]
```

```
{{1, 2, 2}, {0, 1, -1}}
```

In[14]:=
```
Riffle[{a, b, c}, {x, y, z}]
```
Out[14]=
```
{a, x, b, y, c, z}
```

```
PadLeft[{a, b, c}, 10, {x, y, z}, 2]
```
```
{y, z, x, y, z, a, b, c, x, y}
```

```
Tally[Table[RandomInteger[9], {1000}]]
```
```
{{3, 103}, {6, 106}, {4, 91}, {0, 98}, {8, 91}, {9, 98}, {1, 100}, {5, 96}, {2, 118}, {7, 99}}
```

■ Structural operations for general expressions

```
expr = a + b + c
```
```
a + b + c
```

```
Length[expr]
```
```
3
```

```
Append[expr, d]
```
```
a + b + c + d
```

```
Map[Sqrt, expr]
```
$$\sqrt{a} + \sqrt{b} + \sqrt{c}$$

```
expr[[2]]
```
```
b
```

■ Pattern Matching

In[16]:=
```
{f, f[x], f[x, y], f[2]} /. f → g
```
Out[16]=
```
{g, g[x], g[x, y], g[2]}
```

In[18]:=
```
{f, f[x], f[x, y], f[2]} /. f[x] → g[x]
```
Out[18]=
```
{f, g[x], f[x, y], f[2]}
```

In[19]:=
```
{f, f[x], f[x, y], f[2]} /. f[x_] → g[x]
```
Out[19]=
```
{f, g[x], f[x, y], g[2]}
```

In[20]:=
```
{f, f[x], f[x, y], f[2]} /. f[x_Integer] → g[x]
```
Out[20]=
```
{f, f[x], f[x, y], g[2]}
```

```
Cases[Sin[x - 2 y^2], a_[___] → a, {0, Infinity}]
```
```
{Power, Times, Plus, Sin}
```

```
de = D[x^2 * y^2 * f[x, y], x, x, y, y, y]
```
$$12\, f^{(0,1)}[x, y] + 12\, y\, f^{(0,2)}[x, y] + 2\, y^2\, f^{(0,3)}[x, y] + 24\, x\, f^{(1,1)}[x, y] + 24\, x\, y\, f^{(1,2)}[x, y] + 4\, x\, y^2\, f^{(1,3)}[x, y] + 6\, x^2\, f^{(2,1)}[x, y] + 6\, x^2\, y\, f^{(2,2)}[x, y] + x^2\, y^2\, f^{(2,3)}[x, y]$$

```
de /. f[x, y] → 1 /. Derivative[a__][f][x, y] :> (Times @@ ({Dx, Dy}^{a}))
```

$$12\,Dy + 24\,Dx\,Dy\,x + 6\,Dx^2\,Dy\,x^2 + 12\,Dy^2\,y + 24\,Dx\,Dy^2\,x\,y + 6\,Dx^2\,Dy^2\,x^2\,y + 2\,Dy^3\,y^2 + 4\,Dx\,Dy^3\,x\,y^2 + Dx^2\,Dy^3\,x^2\,y^2$$

■ Visualization

```
ListPlot[IntegerExponent[Table[Binomial[200, k], {k, 0, 200}], 2]]
```




```
ListPlot[IntegerExponent[Table[Binomial[200, k], {k, 0, 200}], 2], Joined → True]
```




```
Plot[Table[BesselJ[n, x], {n, 0, 5}], {x, -10, 10}]
```



```
Manipulate[ContourPlot[(x^2 + y^2)^2 == x^2 + a * y^2, {x, -1, 1}, {y, -1, 1}], {a, -1, 1}]
```




```
Plot3D[x^2 / y^2, {x, -1, 1}, {y, -1, 1}]
```



```
Graphics[Table[Circle[{x, 0}, x^2], {x, -2, 2, .1}]]
```



■ Functional Programming

```
f = Function[x, x^2]
```

```
Function[x, x^2]
```

```
f'[3]
```

```
6
```

```
Clear[f]
```

```
D[f[x] * g[x], x]
```

```
g[x] f'[x] + f[x] g'[x]
```

```
Solve[f[x] == y, x]
```

```
InverseFunction::ifun :
 Inverse functions are being used. Values may be lost for multivalued inverses. >>
```

```
{{x → f^(-1)[y]}}
```

```
FullForm[%]
```

```
List[List[Rule[x, InverseFunction[f, 1, 1][y]]]]
```

```
f[InverseFunction[f, 1, 1][x]]
```

```
InverseFunction::ifun :
 Inverse functions are being used. Values may be lost for multivalued inverses. >>
```

```
x
```

In[21]:=
```
FixedPoint[Sin[#] + Cos[#] &, 0.1]
```

Out[21]=
```
1.25873
```

```
Nest[f, x, 5]
```

```
f[f[f[f[f[x]]]]]
```

```
MapIndexed[{#1, #2 - 1} &, CoefficientList[x^3 + y^2 + 2 * x * y, {x, y}], {2}]
```

```
{{{0, {0, 0}}, {0, {0, 1}}, {1, {0, 2}}}, {{0, {1, 0}}, {2, {1, 1}}, {0, {1, 2}}},
 {{0, {2, 0}}, {0, {2, 1}}, {0, {2, 2}}}, {{1, {3, 0}}, {0, {3, 1}}, {0, {3, 2}}}}
```

```
DeleteCases[Flatten[%, 1], {0, _}]
```

```
{{1, {0, 2}}, {2, {1, 1}}, {1, {3, 0}}}
```

Some Exercises

Generate a list of all positive rationals with numerator and denominator not greater than 10!

```
Union[Flatten[Table[n / d, {n, 0, 10}, {d, 10}]]]
```

$$\left\{0, \frac{1}{10}, \frac{1}{9}, \frac{1}{8}, \frac{1}{7}, \frac{1}{6}, \frac{1}{5}, \frac{2}{9}, \frac{1}{4}, \frac{2}{7}, \frac{3}{10}, \frac{1}{3}, \frac{3}{8}, \frac{2}{5}, \frac{3}{7}, \frac{4}{9}, \frac{1}{2}, \frac{5}{9}, \frac{4}{7}, \frac{3}{5}, \frac{5}{8}, \frac{2}{3}, \frac{7}{10}, \frac{5}{7}, \frac{3}{4}, \frac{7}{9}, \frac{4}{5}, \frac{5}{6}, \frac{6}{7}, \frac{7}{8}, \frac{8}{9}, \frac{9}{10}, 1, \frac{10}{9}, \frac{9}{8}, \frac{8}{7}, \frac{7}{6}, \frac{6}{5}, \frac{5}{4}, \frac{9}{7}, \frac{4}{3}, \frac{7}{5}, \frac{10}{7}, \frac{3}{2}, \frac{8}{5}, \frac{5}{3}, \frac{7}{4}, \frac{9}{5}, 2, \frac{9}{4}, \frac{7}{3}, \frac{5}{2}, \frac{8}{3}, 3, \frac{10}{3}, \frac{7}{2}, 4, \frac{9}{2}, 5, 6, 7, 8, 9, 10\right\}$$

All cubes of the smallest totally symmetric plane partition generated by some points:

```
p = {{1, 1, 3}, {2, 2, 2}}
```

```
{{1, 1, 3}, {2, 2, 2}}
```

```
Union[Flatten[
  Table[{a, b, c}, {a, #[[1]]}, {b, #[[2]]}, {c, #[[3]]}] & /@ Flatten[Permutations /@ p, 1], 3]]
```

```
{{1, 1, 1}, {1, 1, 2}, {1, 1, 3}, {1, 2, 1}, {1, 2, 2},
 {1, 3, 1}, {2, 1, 1}, {2, 1, 2}, {2, 2, 1}, {2, 2, 2}, {3, 1, 1}}
```

```
Union[Flatten[
  Table @@@ (Prepend[Transpose[{{a, b, c}, #}], {a, b, c}] & /@ Flatten[Permutations /@ p, 1]), 3]]
```

```
{{1, 1, 1}, {1, 1, 2}, {1, 1, 3}, {1, 2, 1}, {1, 2, 2},
 {1, 3, 1}, {2, 1, 1}, {2, 1, 2}, {2, 2, 1}, {2, 2, 2}, {3, 1, 1}}
```

```
Select[Flatten[With[{m = Max[Flatten[p]]}, Table[{a, b, c}, {a, m}, {b, m}, {c, m}]], 2],
 Function[point, Or @@ ((And @@ Thread[point ≤ #]) & /@ Flatten[Permutations /@ p, 1])]]
```

```
{{1, 1, 1}, {1, 1, 2}, {1, 1, 3}, {1, 2, 1}, {1, 2, 2},
 {1, 3, 1}, {2, 1, 1}, {2, 1, 2}, {2, 2, 1}, {2, 2, 2}, {3, 1, 1}}
```